Lab6: Tables and Figures

Introduction to Econometrics,Fall 2020

Yi Wang

Nanjing University

4/11/2020

Section 1

Basic Graph

Subsection 1

Introduction

Basic Graph : Introduction

- 1.Stata 图形的种类
  - graph twoway 二维图

`scatter`	散点图
`line`	折线图
`area`	区域图
`lfit`	线性拟合图
`histogram`	直方图
`kdensity`	密度函数图
`function`	函数图

  - graph bar 条形图
  - graph box 箱形图
  - graph pie 饼图
  - ... ...

Basic Graph : Introduction

- 2.二维图命令
  - 语法
    - ★ twoway (单元图1) (单元图2) (. . .) ，选项1 选项2 . . .
    - ★ twoway 单元图1 || 单元图2 || . . . ，选项1 选项2 . . .
    - ★ 单元图的定义：(单元图类型 y1 y2 . . . x，选项1 选项2 . . .)
    - ★ 选项的定义：(定义内容，子选项1 子选项2 . . .)
  - 注意：逗号后全部为选项，裸露在外的逗号只有一个

Basic Graph : Introduction

- 2.二维图命令
  - 实例1

```
. sysuse sp500, clear
(S&P 500)
. twoway (line high date) (line low date),                ///
      title("图1: 股票最高价与最低价时序图", box)          ///
      xtitle("交易日期", margin(medsmall))                ///
      ytitle("股票价格")                                  ///
      ylabel(900(200)1400) ymtick(##5)                    ///
      legend(label(1 "最高价") label(2 "最低价"))         ///
      note("资料来源: Stata公司, SP500.dta")              ///
      caption("说明: 我做的第一幅Stata图形! ")            ///
      saving(SP500.gph, replace)
(file SP500.gph saved)


. cap graph export SP500.png, replace
```

Basic Graph : Introduction

- 2.二维图命令
  - 实例1



Basic Graph : Introduction

- 2.二维图命令
  - 实例2

```
. twoway (function y=sin(x), range(-20 20) lw(*1.5))  ///
         (function y=cos(x), range(-20 20) lw(*2.0)), ///
         ytick(-2(0.5)2) ylabel(, angle(0))           ///
         yline(0, lcolor(black*0.2) lpattern(dash))   ///
         scheme(s1rcolor )                            ///
         saving(trigono.gph, replace)


. cap graph export trigono.png, replace
```

Basic Graph : Introduction

- 2.二维图命令
  - 实例2



Basic Graph : Introduction

- 2.二维图命令
  - 图形无非是点、线(面)、文字等元素的组合
  - 这些组合的整体“风格”构成了图类：单元图(,前的部分)
  - 每种图形的具体特征由元素的特征决定：选项(,后的部分)
  - 因此，选项的填写是Stata绘图的关键!

Subsection 2

Management

Basic Graph : Management

- 1.保存
  - 【方法一】 *-graph save-*

```
sysuse sp500, clear
twoway line high date
graph save fig1.gph, replace
```

  - 【方法二】 *saving(filename, . . .)*

```
twoway line high date, saving(fig11.gph, replace)
```

  - 【方法三】 右击 —> *Save graph . . .*

Basic Graph : Management

- 2.导出
  - 可选择的图形格式：.ps .eps .svg .wmf .emf .pdf .png .tif .gif .jpg . . .

```
graph export fig1.wmf, replace

graph export "D:\Teaching\Stata\lab6\fig1.wmf", replace

*调整图片的分辨率(适用于.png 和 .tif )
graph export fig1.tif, width(800) height(600) replace
```

Basic Graph : Management

- 3.调入

```
graph use trigono.gph
graph use trigono, scheme(s1mono)
```

- 4.显示

```
graph display, scheme(sj)
```

Basic Graph : Management

- 5.查询

```
graph dir
```

- 6.删除

```
erase trigono.gph
graph drop _all
```

Basic Graph : Management

- 7.合并

```
. help graph combine
```

```
. sysuse sp500, clear
(S&P 500)
. twoway line high low date
.   graph save fig2.gph, replace
(file fig2.gph saved)
. twoway line high low date, scheme(s1mono)
.   graph save fig3.gph, replace
(file fig3.gph saved)
. graph combine fig2.gph fig3.gph , saving(sp500_c,replace)
(file sp500_c.gph saved)


. cap graph export sp500_c.png, replace
```

Basic Graph : Management

- 7.合并



Basic Graph : Management

- 8.显示模式
  - Stata提供的模板：

```
help schemes
```

  - stata 用户提供的模板：
    - ★ Mitchell, M.
    - ★ A Visual Guide to Stata Graphics, Third Edition

```
net from http://www.stata-press.com/data/vgsg2/
net install vgsg2   // 安装外部模式插件
net get vgsg2       // 下载相关数据
```

Basic Graph : Management

- 8.显示模式
  - e.g.

```
. use allstates.dta, clear
. scatter propval100 rent700 ownhome, scheme(vg_outm)
.     cap graph export vg1.png, replace
```



Basic Graph： Management

- 8.显示模式
  - e.g.

```
. scatter propval100 rent700 ownhome, scheme(vg_outc)
.     cap graph export vg1.png, replace
```



Basic Graph : Management

- 8.显示模式
  - 其他模板

```
findit scheme
```

Subsection 3

title_options

Basic Graph : title_options

- 标题种类：主标题、副标题、注释、说明
- *title()*、*subtitle()*、*note()*、*caption()*

```
help title_options
```

Basic Graph : title_options

- 1.主标题

```
. use allstates.dta, clear
(Data on 50 States)
. scatter propval100 ownhome, title("My title")

. cap graph export ti_1.png, replace
```



Basic Graph : title_options

- 2.副标题

```
. scatter propval100 ownhome, title("My title") subtitle("My subtitle")

. cap graph export ti_2.png, replace
```



Basic Graph : title_options

- 3.说明和注释

```
. scatter propval100 ownhome, title("My title") subtitle("My subtitle")   ///
  caption("My caption") note("My note")

. cap graph export ti_3.png, replace
```



Basic Graph : title_options

- 4.标题的位置

```
. scatter propval100 ownhome, title("My title", position(6))

. cap graph export ti_4.png, replace
```



Basic Graph : title_options

- 4.标题的位置

```
. scatter propval100 ownhome, title("My title", position(1) ring(0))


. cap graph export ti_5.png, replace


* ring(0) = 绘图区内
* ring(k), k>0, 绘图区以外
* ring()的值越大, 距离绘图区越远
```



Basic Graph : title_options

```
. scatter propval100 ownhome, title("This is my" "title",    ///
  box bexpand justification(left) span)


. cap graph export ti_6.png, replace
```

*box : add a box
*bexpand : fill the width of the plot region
*ustification(left) : left justify the text inside the box
*span : make the box span the entire width of the graph



Subsection 4

axis_options

Basic Graph : axis_options

- 1.坐标轴刻度(tick)及刻度标签(label)

```
help axis_label_options
```

  - 主刻度及标签：*ylabel(), xlabel()*—-major ticks plus labels
  - 主刻度： *ytick(), xtick()*—-major ticks only
  - 子刻度及标签：*ymlabel(), xmlabel()*—-minor ticks plus labels
  - 子刻度： *ymtick(), xmtick()*—-minor ticks only

Basic Graph : axis_options

- 1.坐标轴刻度(tick)及刻度标签(label)

```
. twoway scatter propval100 faminc, xlabel(#10) ylabel(#10)

. cap graph export ax_1.png, replace

* ask for about 10 values to be labeled on each axis
```



Basic Graph : axis_options

- 1.坐标轴刻度(tick)及刻度标签(label)

```
twoway scatter propval100 faminc, ylabel(0(10)100)
twoway scatter propval100 faminc, ymlabel(10(20)90)
twoway scatter propval100 faminc, ytick(#10)
twoway scatter propval100 faminc, ymtick(##10)
```

```
. twoway scatter propval100 faminc, xlabel(,angle(45)) ylabel(,angle(-15))
. cap graph export ax_2.png, replace
```



Basic Graph : axis_options

- 1.坐标轴刻度(tick)及刻度标签(label)

```
. twoway scatter propval100 region, xlabel(1 "NorthEast" 2  ///
      "NorthCentral" 3 "Southern" 4 "Western")


. cap graph export ax_3.png, replace
```



Basic Graph : axis_options

- 2.坐标轴标题: ytitle() xtitle()

```
help axis_label_options
```

Basic Graph : axis_options

- 2.坐标轴标题: ytitle() xtitle()

```
. twoway scatter ownhome propval100,                                ///
      ytitle("Percent of households that own their homes")  ///
      xtitle("Percent of homes over $100,000", size(small) color(navy) place(right))


. cap graph export ax_4.png, replace
```



Basic Graph : axis_options

- 3.坐标结构: yscale() xscale()

```
help axis_scale_options
```

Basic Graph : axis_options

- 3.坐标结构: yscale() xscale()

```
. twoway scatter educ popden, xscale(log)
. cap graph export ax_5.png, replace

* the axis should be displayed on a log scale
```



Basic Graph : axis_options

- 3.坐标结构: yscale() xscale()

```
. twoway scatter educ popden, yscale(noline) xscale(noline)
. cap graph export ax_6.png, replace
```

*不显示坐标轴



Basic Graph : axis_options

- 3.坐标结构: yscale() xscale()

```
. twoway scatter educ popden, xscale(range(-1000 11000))
. cap graph export ax_7.png, replace
```

* 显示范围



Basic Graph : axis_options

- 3.坐标结构: yscale() xscale()

```
. twoway scatter educ popden, xscale(lcolor(red) lwidth(vthick))

. cap graph export ax_8.png, replace

* 坐标轴线型
```



Basic Graph : axis_options

- 4.双坐标系

```
help axis_scale_options
```

Basic Graph : axis_options

- 4.双坐标系

```
* 共用 x 轴

. twoway (scatter faminc educ, yaxis(1))    ///
         (scatter workers2 educ, yaxis(2))
. cap graph export ax_9.png, replace
```



Basic Graph : axis_options

- 4.双坐标系

```
. twoway (scatter faminc educ)                    ///
         (scatter workers2 educ, yaxis(2)),       ///
         ylabel(40(5)80, axis(2))

. cap graph export ax_10.png, replace

* Without the axis(2) option, Stata would assume that we are referring to the first axi
```



Subsection 5

legend_options

Basic Graph : legend_options

- 一张图中同时呈现多个序列，便会自动产生图例
- 对于变量而言，其默认图例是它的变量标签

```
help legend_options
```

Basic Graph : legend_options

```
. set scheme vg_s2c

. twoway (scatter ownhome urban) (lfit ownhome urban) (qfit ownhome urban)

. cap graph export le_1.png, replace
```



Basic Graph : legend_options

```
. twoway (scatter ownhome urban) (lfit ownhome urban) (qfit ownhome urban),  ///
      legend(order(2 3 1))

. cap graph export le_2.png, replace
```



Basic Graph : legend_options

```
. twoway (scatter ownhome urban) (lfit ownhome urban) (qfit ownhome urban),  ///
      legend(label(1 "% Own home") label(2 "Lin.Fit") label(3 "Quad. Fit"))

. cap graph export le_3.png, replace
```



Basic Graph : legend_options

```
. twoway (scatter ownhome urban) (lfit ownhome urban) (qfit ownhome urban),  ///
      legend(order(1 "% Own Home" 2 "Linear" 3 "Quad") rows(1) position(12))

. cap graph export le_4.png, replace
```



Basic Graph : legend_options

```
. twoway (scatter ownhome urban) (lfit ownhome urban) (qfit ownhome urban),  ///
    legend(rows(1) title("Legend", color(red) size(huge) box bexpand))

. cap graph export le_5.png, replace
```



Basic Graph : legend_options

```
. twoway (scatter ownhome urban) (lfit ownhome urban) (qfit ownhome urban),  ///
      legend(size(large) color(maroon) fcolor(eggshell) box)

. cap graph export le_6.png, replace
```



Subsection 6

region_options

Basic Graph : region_options

- Stata图形的区域划分

```
help axis_label_options
```

Basic Graph : region_options

- Stata图形的区域划分

```
. sysuse auto, clear
(1978 Automobile Data)

. scatter mpg weight,                                          ///
    graphregion(fcolor(lavender*0.8))                          ///
    graphregion(ifcolor(eggshell))                             ///
    plotregion(fcolor(black*0.3))                              ///
    plotregion(ifcolor(white))                                 ///
    title("全图内区 (inner graph region)")                      ///
    xtitle("汽车重量")                                          ///
    ytitle("汽车里数")                                          ///
    note("(1) 灰色部分--“绘图外区”(outer plot region)"          ///
         "(2) 紫色部分--“全图外区”(outer graph region)")        ///
    caption("各区域的面积决定于显示模式(scheme)和内容的多少") ///
    text(30 3500 "绘图内区 (inner plot region)", color(blue) size(*1.5))
```

Basic Graph： region_options

- Stata图形的区域划分

```
. cap graph export re_1.png, replace
```



Subsection 7

added_line_options

Basic Graph : added_line_options

- 附加线

```
help added_line_options
```

  - 语法

```
twoway ..., yline(数字, 子选项) xline(数字, 子选项)
```

  - 数字：控制附加线的位置
  - 子选项：控制附加线的类型、颜色、宽度等

Basic Graph : added_line_options

- 附加线

```
. sysuse sp500.dta,clear

. line open date,                                                          ///
      yline(1100,lwidth(*1.5) lpattern(shortdash_dot) lcolor(blue*0.6))    ///
      yline(1313,lw(*2.5) lc(green*0.4))                                   ///
      xline(15242,lw(*2) lc(pink*0.4) lp(longdash))

. cap graph export al_1.png, replace
```



Subsection 8

added_text_options

Basic Graph : added_text_options

- 文字与文本框

```
help textbox_options
```

  - 凡是出现文字的地方都可以做下面的设定
  - 文字和文本框的整体风格: 标题、副标题、文本、小号
  - 文本框相关设定: 文本框颜色、背景、与文字的边距等
  - 文字相关的设定: 大小、颜色、位置、行距等

Basic Graph : added_text_options

- 文字与文本框

```
. line open date, text(1324.83 15117 "一个波峰",color(blue) size(*1.6))

. cap graph export at_1.png, replace
```



Subsection 9

marker_options & marker_label_options

Basic Graph : marker_options & marker_label_options

- 图标和图标的标签

```
help marker_options
help marker_label_options
```

Basic Graph : marker_options & marker_label_options

- 图标和图标的标签

```
. use allstates.dta, clear
(Data on 50 States)
. twoway scatter ownhome borninstate, msymbol(X) msize(small)

. cap graph export ma_1.png, replace
```



Basic Graph : marker_options & marker_label_options

- 图标和图标的标签

```
. twoway scatter ownhome borninstate [aweight=propval100],    ///
      msymbol(oh) msize(small)

. cap graph export ma_2.png, replace
```



Basic Graph : marker_options & marker_label_options

- 图标和图标的标签

```
. twoway scatter ownhome borninstate [aweight=propval100],    ///
      msymbol(oh) msize(large) mlabel(stateab)

. cap graph export ma_3.png, replace
* the weights no longer affect the size of the markers
```



Basic Graph : marker_options & marker_label_options

- 图标和图标的标签

```
. twoway (scatter ownhome borninstate [aweight=propval100], msymbol(oh) msize(large)) /
      (scatter ownhome borninstate, mlabel(stateab) msymbol(i) mlabpos(center))

. cap graph export ma_4.png, replace


/* To solve:overlaying a scatterplot that has the symbols weighted by propval100
  with a scatterplot that shows just the marker labels.*/
```



Basic Graph : marker_options & marker_label_options

- 图标和图标的标签

```
. twoway scatter ownhome borninstate, mfcolor(ltblue) mlcolor(navy) mlwidth(medthick)

. cap graph export ma_5.png, replace
```



Basic Graph : marker_options & marker_label_options

- 图标和图标的标签

```
. twoway scatter ownhome borninstate,   ///
  mlabel(stateab) mlabpos(12) mlabsize(vsmall) mlabangle(45) mlabcolor(red)

. cap graph export ma_6.png, replace
```



Subsection 10

by_option

Basic Graph : by_option

- 分组绘图

```
help by_option
```

Basic Graph : by_option

- 分组绘图

```
. use allstates.dta, clear
(Data on 50 States)
. twoway scatter ownhome borninstate, by(north)
. cap graph export by_1.png, replace
```



Basic Graph : by_option

- 分组绘图

```
. twoway scatter ownhome borninstate, by(north, total)
. cap graph export by_2.png, replace
```



Basic Graph : by_option

- 分组绘图

```
. twoway scatter ownhome borninstate, by(north, total holes(2))
. cap graph export by_3.png, replace
```



Basic Graph : by_option

- 分组绘图

```
. twoway scatter ownhome borninstate, by(north, total rows(1))
. cap graph export by_4.png, replace
```



Basic Graph : by_option

- 分组绘图

```
. twoway scatter ownhome borninstate, by(north, total iscale(*1.5))
. cap graph export by_5.png, replace
```



Basic Graph : by_option

- 分组绘图

```
. twoway scatter ownhome borninstate, by(north, rescale)
. cap graph export by_6.png, replace
```

```
* scales both the variable and variable differently across the by-groups. Both axes ar
  arately rescaled.
```



Basic Graph : by_option

- 分组绘图

```
. twoway scatter ownhome borninstate, by(north, title("By title")) title("Regular title
. cap graph export by_7.png, replace
```



Basic Graph : by_option

- 分组绘图

```
. twoway scatter (borninstate propval100 ownhome),   ///
>     by(nsw,legend(position(12))) legend(label(1 "Born in state") label(2 "% > 100K"))
. cap graph export by_8.png, replace
```



Subsection 11

linear prediction plots

Basic Graph : linear prediction plots

```
help twoway lfit
help twoway qfit
```

- 简单示例

```
. scatter ownhome pcturban80 || lfit ownhome pcturban80, by(north, total row(1))
. cap graph export lp_1.png, replace
```



Basic Graph : linear prediction plots

```
help twoway lfitci
help twoway qfitci
```

- 附加置信区间

```
* 线性拟合的置信区间
. twoway (lfitci ownhome pcturban80, stdf) (scatter ownhome pcturban80)
. cap graph export lp_2.png, replace
```



Basic Graph : linear prediction plots

```
* 非线性拟合
. twoway (qfitci ownhome pcturban80, stdf) (scatter ownhome pcturban80)

. cap graph export lp_3.png, replace
```



Basic Graph : linear prediction plots

```
. twoway (lfitci ownhome pcturban80,                              ///
ciplot(rline) lcolor(green) lpattern(dash) lwidth(thick))   ///
      (scatter ownhome pcturban80)
. cap graph export lp_4.png, replace
```



Basic Graph : linear prediction plots

```
. twoway (lfitci ownhome pcturban80, stdf level(99) color(eggshell)) ///
         (lfitci ownhome pcturban80, stdf level(90) color(lavender)) ///
         (scatter ownhome pcturban80, mcolor(black))

. cap graph export lp_5.png, replace
```



Section 2

Nonlinear Regression

Subsection 1

Review the Theory

Nonlinear Regression

- Review the Theory

Nonlinear in Xs

Nonlinear Regression Regression Functions

- So far our regression model is

$$Y_i = \beta_0 + \beta_1 X_{1,i} + ... + \beta_k X_{k,i} + u_i$$

- The effect of Y on a change in X_j by 1 (unit) is constant and equals β_j:

$$\beta_j = \frac{\partial Y_i}{\partial X_{ji}}$$

- But if a relation between Y and X is nonlinear:
  - The effect on Y of a change in X depends on the value of X – that is, the *marginal effect of X is not constant*.
  - A linear regression is misspecified – the functional form is wrong.
  - The estimator of the effect on Y of X is biased(a special case of OVB).
- The solution to this is to estimate a regression function that is nonlinear in X.

Nonlinear Regression

- Review the Theory

Nonlinear in Xs

Two Cases:

1. The effect of change in X_1 on Y depends on X_1 itself.
   - eg. the effect of a change in class size on test scores is bigger when initial class size is small.
2. The effect of change in X_1 on Y depends on another variable, like X_2
   - eg.the effect of class size on test scores depends on the percentage of disadvantaged pupils in the class.

Nonlinear Regression

- Review the Theory

Nonlinear in Xs

Two Complementary Approaches:

1. **Polynomials in X**
   - The population regression function is approximated by a quadratic, cubic, or higher-degree polynomial.
2. **Logarithmic transformations**
   - Y and/or X is transformed by taking its logarithm
   - this gives a *percentages* interpretation that makes sense in many applications

Nonlinear Regression

- Review the Theory

Interactions Between Independent Variables

Introduction

- The product of two variables is called an **interaction term**.
- Try to answer *how the effect on Y of a change in an independent variable depends on the value of another independent variable.*
- Consider three cases:

1. Interactions between two binary variables.
2. Interactions between a binary and a continuous variable.
3. Interactions between two continuous variables.

Subsection 2

In practice

Nonlinear Regression

- In practice

```
. use caschool.dta, clear
```

```
. ***** 数据转化 ************

. gen avginc2 = avginc^2    //avginc平方项
. gen avginc3 = avginc^3    //avginc立方项

. gen loginc = ln(avginc)   //avginc对数
. gen logtest = ln(testscr) //testscr对数

. gen loginc2 = loginc^2    //avginc对数平方项
. gen loginc3 = loginc^3    //avginc对数立方项

. gen histr = (str>=20)     //histr = 1 if the str > 20
. gen hiel = (el_pct >= 10) //hiel = 1 if the el_pct > 10%
```

Nonlinear Regression

- In practice

```
. gen hisxhie = histr*hiel  //D1i × D2i
. gen strxhiel = str*hiel      //Xi × Di
. gen strxelpc = str*el_pct //X1i × X2i

. gen sttr2 = str^2         //str平方项
. gen sttr3 = str^3           //str立方项

. gen str2hiel = sttr2*hiel //Xi × Di
. gen str3hiel = sttr3*hiel //Xi × Di
```

Nonlinear Regression

- In practice

$$\text{Test Score} = \beta_0 + \beta_1 STR + \beta_2 \text{ English} + \beta_3 \text{ Lunch} + u$$

```
. reg testscr str el_pct meal_pct, r    //多元线性回归模型
Linear regression                               Number of obs     =        420
                                                F(3, 416)         =     453.48
                                                Prob > F          =     0.0000
                                                R-squared         =     0.7745
                                                Root MSE          =     9.0801
```

testscr	Coef.	Robust Std. Err.	t	P>|t|	[95% Conf.	Interval]
str	-.9983092	.2700799	-3.70	0.000	-1.529201	-.4674178
el_pct	-.1215733	.0328317	-3.70	0.000	-.18611	-.0570366
meal_pct	-.5473456	.0241072	-22.70	0.000	-.5947328	-.4999583
_cons	700.15	5.56845	125.74	0.000	689.2042	711.0958

```
. est store m1
```

Nonlinear Regression

- In practice

$$\text{Test Score} = \beta_0 + \beta_1 STR + \beta_2 \text{ English} + \beta_3 \text{ Lunch} + \beta_4 \ln(\text{Income}) + u$$

```
. reg testscr str el_pct meal_pct loginc, r //控制对数收入的影响
Linear regression                               Number of obs     =        420
                                                F(4, 415)         =     417.20
                                                Prob > F          =     0.0000
                                                R-squared         =     0.7962
                                                Root MSE          =     8.6426
```

testscr	Coef.	Robust Std. Err.	t	P>|t|	[95% Conf.	Interval]
str	-.734326	.2567803	-2.86	0.004	-1.239078	-.2295738
el_pct	-.1755344	.0336606	-5.21	0.000	-.241701	-.1093678
meal_pct	-.3982342	.0331741	-12.00	0.000	-.4634443	-.333024
loginc	11.56897	1.818811	6.36	0.000	7.993736	15.1442
_cons	658.552	8.641528	76.21	0.000	641.5653	675.5386

```
. est store m2
```

Nonlinear Regression

- In practice

$$\text{Test Score} = \beta_0 + \beta_1 \text{ STR } + \beta_2 \text{ HiEL} + \beta_3 (STR \times HiEL) + u$$

```
. reg testscr str hiel strxhiel, r //考虑生师比与英语学习者百分比的交互影响
Linear regression                               Number of obs     =        420
                                                F(3, 416)         =      63.67
                                                Prob > F          =     0.0000
                                                R-squared         =     0.3103
                                                Root MSE          =      15.88
------------------------------------------------------------------------------
             |               Robust
     testscr |      Coef.   Std. Err.      t    P>|t|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
         str |   -.9684601   .5891016    -1.64   0.101    -2.126447    .1895268
        hiel |    5.639141   19.51456     0.29   0.773    -32.72029    43.99857
    strxhiel |   -1.276613   .9669194    -1.32   0.187     -3.17727    .6240436
       _cons |    682.2458   11.86781    57.49   0.000     658.9175    705.5742
------------------------------------------------------------------------------
. est store m3
```

Nonlinear Regression

- In practice

```
. test str strxhiel //联合假设的F 统计量和P 值
 ( 1)  str = 0
 ( 2)  strxhiel = 0
       F(  2,   416) =    5.64
            Prob > F =    0.0038
```

Nonlinear Regression

- In practice

$$\text{Test Score} = \beta_0 + \beta_1 \text{ STR } + \beta_2 \text{HiEL} + \beta_3 (STR \times HiEL)$$
$$+\beta_4 \text{ Lunch } + \beta_5 \ln(\text{Income}) + u$$

```
. reg testscr str hiel strxhiel meal_pct loginc, r //控制学生特征的基础上考虑交互影响
Linear regression                               Number of obs     =        420
                                                F(5, 414)         =     335.85
                                                Prob > F          =     0.0000
                                                R-squared         =     0.7974
                                                Root MSE          =     8.6286
------------------------------------------------------------------------------
             |               Robust
    testscr |      Coef.   Std. Err.      t    P>|t|     [95% Conf. Interval]
------------+----------------------------------------------------------------
         str |  -.5310319   .3418468    -1.55   0.121    -1.203004      .14094
        hiel |   5.498208      9.795     0.56   0.575    -13.75593    24.75234
    strxhiel |  -.5776664   .4957779    -1.17   0.245    -1.552222    .3968896
    meal_pct |  -.4113776   .0288276   -14.27   0.000    -.4680443   -.3547109
      loginc |   12.12447   1.797513     6.75   0.000     8.591078    15.65786
       _cons |   653.6661   9.869378    66.23   0.000     634.2658    673.0665
------------------------------------------------------------------------------
. est store m4
```

Nonlinear Regression

- In practice

```
. test str strxhiel //联合假设的F统计量和P值
 ( 1)  str = 0
 ( 2)  strxhiel = 0
       F(  2,   414) =    5.92
            Prob > F =    0.0029
```

Nonlinear Regression

- In practice

$$\text{TestScore} = \beta_0 + \beta_1 \times \text{STR} + \beta_2 \times \text{STR}^2 + \beta_3 \times \text{STR}^3$$
$$+\beta_4 \text{ HiEL} + \beta_5 \text{ Lunch} + \beta_6 \ln(\text{Income}) + u$$

```
. reg testscr str sttr2 sttr3 hiel meal_pct loginc, r //考虑生师比的非线性影响
Linear regression                               Number of obs     =        420
                                                F(6, 413)         =     281.14
                                                Prob > F          =     0.0000
                                                R-squared         =     0.8011
                                                Root MSE          =     8.5593
------------------------------------------------------------------------------
             |               Robust
     testscr |      Coef.   Std. Err.      t    P>|t|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
         str |   64.33964   24.86053     2.59   0.010     15.47069    113.2086
       sttr2 |  -3.423925   1.249901    -2.74   0.006    -5.880886   -.9669636
       sttr3 |   .0592895    .020763     2.86   0.005     .0184752    .1001037
        hiel |  -5.473991   1.033762    -5.30   0.000    -7.506081     -3.4419
    meal_pct |  -.4200571   .0285258   -14.73   0.000    -.4761309   -.3639833
      loginc |   11.74819   1.771437     6.63   0.000     8.266032    15.23035
       _cons |   252.0458   163.6341     1.54   0.124    -69.61384    573.7055
------------------------------------------------------------------------------
. est store m5
```

Nonlinear Regression

- In practice

```
. test str sttr2 sttr3   //联合假设的F统计量和P值
 ( 1)  str = 0
 ( 2)  sttr2 = 0
 ( 3)  sttr3 = 0
       F(  3,   413) =    6.31
            Prob > F =    0.0003
. test sttr2 sttr3       //联合假设的F统计量和P值
 ( 1)  sttr2 = 0
 ( 2)  sttr3 = 0
       F(  2,   413) =    6.17
            Prob > F =    0.0023
```

Nonlinear Regression

- In practice

$$\begin{aligned}\text{TestScore} &= \beta_0 + \beta_1 \times \text{STR} + \beta_2 \times \text{STR}^2 + \beta_3 \times \text{STR}^3 \\ &+ \beta_4\ \text{HiEL} + \beta_5\,(STR \times HiEL) + \beta_6\left(STR^2 \times HiEL\right) + \\ &\beta_7\left(STR^3 \times HiEL\right) + \beta_8\ \text{Lunch}\ + \beta_9 \ln(\text{Income}) + u\end{aligned}$$

```
. reg testscr str sttr2 sttr3 hiel strxhiel str2hiel str3hiel meal_pct loginc, r
//考虑所有因素
Linear regression                               Number of obs     =        420
                                                F(9, 410)         =     199.88
                                                Prob > F          =     0.0000
                                                R-squared         =     0.8031
                                                Root MSE          =     8.5475
------------------------------------------------------------------------------
             |               Robust
     testscr |      Coef.   Std. Err.      t    P>|t|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
         str |   83.70272    28.4967     2.94   0.003     27.68485    139.7206
       sttr2 |  -4.380841   1.441017    -3.04   0.003    -7.213544   -1.548139
       sttr3 |   .0749236    .024008     3.12   0.002     .0277295    .1221178
        hiel |   816.0904   327.6745     2.49   0.013     171.9587    1460.222
    strxhiel |  -123.2842   50.21281    -2.46   0.014    -221.9909   -24.57757
    str2hiel |    6.12124   2.541978     2.41   0.016     1.124303    11.11818
    str3hiel |     -.1006   .0425094    -2.37   0.018    -.1841636   -.0170364
    meal_pct |  -.4177876   .0287011   -14.56   0.000    -.4742073   -.3613679
      loginc |   11.80035   1.778008     6.64   0.000     8.305197    15.2955
       _cons |   122.3464    185.518     0.66   0.510    -242.3388    487.0315
------------------------------------------------------------------------------
. est store m6
```

Nonlinear Regression

- In practice

```
. test str sttr2 sttr3 strxhiel str2hiel str3hiel //联合假设的F 统计量和P 值
 ( 1)  str = 0
 ( 2)  sttr2 = 0
 ( 3)  sttr3 = 0
 ( 4)  strxhiel = 0
 ( 5)  str2hiel = 0
 ( 6)  str3hiel = 0
       F(  6,   410) =    4.96
            Prob > F =    0.0001
```

Nonlinear Regression

- In practice

```
. test sttr2 sttr3 //联合假设的F 统计量和P 值
 ( 1)  sttr2 = 0
 ( 2)  sttr3 = 0
       F(  2,   410) =    5.81
            Prob > F =    0.0033
```

Nonlinear Regression

- In practice

```
. test strxhiel str2hiel str3hiel //联合假设的F 统计量和P 值
 ( 1)  strxhiel = 0
 ( 2)  str2hiel = 0
 ( 3)  str3hiel = 0
       F(  3,   410) =    2.69
            Prob > F =    0.0460
```

Nonlinear Regression

- In practice

$$\text{TestScore} = \beta_0 + \beta_1 \times \text{STR} + \beta_2 \times \text{STR}^2 + \beta_3 \times \text{STR}^3$$
$$+\beta_4\ \text{English} + \beta_5\ \text{Lunch} + \beta_6 \ln(\text{Income}) + u$$

```
. reg testscr str sttr2 sttr3 el_pct meal_pct loginc, r //除交互影响外所有因素
Linear regression                               Number of obs     =        420
                                                F(6, 413)         =     280.81
                                                Prob > F          =     0.0000
                                                R-squared         =     0.8007
                                                Root MSE          =     8.5679
```

testscr	Coef.	Robust Std. Err.	t	P>|t|	[95% Conf.	Interval]
str	65.28595	25.25864	2.58	0.010	15.63443	114.9375
sttr2	-3.465567	1.270893	-2.73	0.007	-5.963793	-.9673414
sttr3	.059909	.0211205	2.84	0.005	.0183919	.1014262
el_pct	-.165687	.0343657	-4.82	0.000	-.2332405	-.0981334
meal_pct	-.4024177	.0332667	-12.10	0.000	-.4678108	-.3370246
loginc	11.50894	1.806403	6.37	0.000	7.958044	15.05983
_cons	244.8025	165.7221	1.48	0.140	-80.9614	570.5665

```
. est store m7
```

Nonlinear Regression

- In practice

```
. test str sttr2 sttr3 //联合假设的F 统计量和P 值
 ( 1)  str = 0
 ( 2)  sttr2 = 0
 ( 3)  sttr3 = 0
       F(  3,   413) =    5.91
            Prob > F =    0.0006
```

Nonlinear Regression

- In practice

```
. test sttr2 sttr3 //联合假设的F 统计量和P 值
 ( 1)  sttr2 = 0
 ( 2)  sttr3 = 0
       F(  2,   413) =    5.96
            Prob > F =    0.0028
```

Nonlinear Regression

- In practice

```
*导出到Latex（或Word）
. esttab m* using nl_ols.tex, replace                    ///
      star( * 0.10 ** 0.05 *** 0.01 ) compress           ///
      b(%6.3f) t(%6.3f) r2(%9.3f) ar2                   ///
      booktabs page width(\\hsize)
(output written to nl_ols.tex)
```

Nonlinear Regression

- In practice

	(1) testscr	(2) testscr	(3) testscr	(4) testscr	(5) testscr	(6) testscr	(7) testscr
str	-0.998***	-0.734***	-0.968	-0.531	64.340***	83.703***	65.286**
	(-3.696)	(-2.860)	(-1.644)	(-1.553)	(2.588)	(2.937)	(2.585)
el_pct	-0.122***	-0.176***					-0.166***
	(-3.703)	(-5.215)					(-4.821)
meal_pct	-0.547***	-0.398***		-0.411***	-0.420***	-0.418***	-0.402***
	(-22.705)	(-12.004)		(-14.270)	(-14.726)	(-14.556)	(-12.097)
loginc		11.569***		12.124***	11.748***	11.800***	11.509***
		(6.361)		(6.745)	(6.632)	(6.637)	(6.371)
hiel			5.639	5.498	-5.474***	816.090**	
			(0.289)	(0.561)	(-5.295)	(2.491)	
strxhiel			-1.277	-0.578		-123.284**	
			(-1.320)	(-1.165)		(-2.455)	
sttr2					-3.424***	-4.381***	-3.466***
					(-2.739)	(-3.040)	(-2.727)
sttr3					0.059***	0.075***	0.060***
					(2.856)	(3.121)	(2.837)
str2hiel						6.121**	
						(2.408)	
str3hiel						-0.101**	
						(-2.367)	
_cons	700.150***	658.552***	682.246***	653.666***	252.046	122.346	244.803
	(125.735)	(76.208)	(57.487)	(66.232)	(1.540)	(0.659)	(1.477)
N	420	420	420	420	420	420	420
R^2	0.775	0.796	0.310	0.797	0.801	0.803	0.801
adj. R^2	0.773	0.794	0.305	0.795	0.798	0.799	0.798